VBA Code Sample #38 ( The Outlook Automation Access 2002 Add-in

Created 12-25-2001
Helen Feddema (hfeddema@hvc.rr.com)

Instructions

The Outlook Automation add-in consists of a single menu add-in that runs a function that imports data from standard or custom Contact items to an Access table in the current database.  Additionally (though it has nothing to do with Outlook Automation), the add-in includes a function (another menu add-in) that produces a report listing all the fields in all the tables of the current database.  The add-in works in Office XP (2002); a version of this add-in that works in Office 97 is available as Code Sample #15, and an Office 2000 version as Code Sample #18.

To install the add-in, copy the files in the zip to the locations indicated in Table 1 below:

Table 1:  File Locations

	File Name
	Install to Folder

	Outlook Automation.mda
	Add-ins folder (usually C:\WINDOWS\Application Data\Microsoft\AddIns for Windows ME or C:\Documents and Settings\UserName\Application Data\Microsoft\AddIns for Windows 2000)

	OA Test Form.oft
	\Templates\Outlook

	*.msg
	\My Documents\ (or wherever you want)


Open any database, drop down the Tools menu, and select Add-In Manager from the fly-out menu:

If you see an Outlook Automation selection, just select it and click the Install button, as shown below; if you don't see it, click the Add New button and select Outlook Automation.mda from its location in the Add-ins folder (or wherever else you may have installed it).

After closing the Add-in Manager dialog, you should see two new selections on the Add-ins menu:

Get Contact Data

To test the Outlook Automation add-in, follow the steps below.

1. Create a folder called "Contacts to Export" (or whatever you wish) in your Outlook folders system.

2. Drag the *.msg files to this folder to create Contact items.  (These items are based on a custom Outlook contact form called “OA Test Form.”)

3. If desired, copy other contact items into this folder, or create new ones, based on the standard Contact form or a custom contact form.

4. Open one of the contacts from the zip file (say Ann Jaszynski).

5. Open any Access 2002 database and check that it has references set to the current versions of Outlook and DAO.  To do this, open the VBE window, then select References from the Tools dialog.  If the DAO and Outlook items are not checked, locate them and check them (DAO could be either 3.51 or 3.60).

6. Close the References dialog, and select the Get Contact Data command from the Add-Ins menu.

The code first verifies that Outlook is open, a contact item is open, and the current Outlook folder contains contact items.  Next, you will then be asked whether you want to import all the contact items in the folder, or just items based on the open form.  The open form tells the code what form to filter by, when processing the contact items in the current folder.

Next, you will get the obnoxious Object Model Guardian dialog (a new and unwelcome feature – or perhaps more accurately, obstacle – of Outlook 2002).  Check the Allow access for checkbox, and select 10 minutes for the maximum time allowed, then click the Yes button.

If you have a standard form open, or a custom form that doesn’t contain any custom fields (say, just built-in fields that don’t appear on the standard Contact form, or the usual controls arranged in a different order), the data will be imported and the table of data (tblOutlookContacts) will be opened automatically:

In case you have a custom Contact form with custom fields open, the process is a little different; the code creates a new module in your database, called basGetCustomContactData, which has code designed specifically for the Outlook form you have open, with a function (ImportCustomContactData) which does the import.  Since the procedure in the module just created from code can't be run from the same code, I created a macro to run the function for convenience (mcrImportCustomContactData).  When the code ends, you will get the following message:

You can examine the basGetCustomContactData module and (if necessary) edit it, say to remove fields you aren’t actually using (you may be surprised to see how many custom fields there are in an item!).  Any time you want to run the same import again, you can just run the mcrImportCustomContactData macro to re-run the import function.  When you run the macro, the function will check that the currently open Contact item matches the one the module was designed for, and will advise you to run the Get Contact Data command from the menu again if you have a contact item based on a different form open.  You will also be offered a choice of clearing tblOutlookContacts and starting fresh, or appending the new batch of contacts to the table.

Unfortunately, because of the lack of a unique Contact ID in Outlook, there is no way to update existing records, so if you choose to append, and you are processing the same folder as before, you may end up with duplicate records.

After running this macro, tblOutlookContacts will be filled with data from the custom forms, including custom fields.  The sample table below shows data in some built-in fields that don't appear on the standard Contact form, as well as data in custom fields:

[image: image8.png]&
Ix

Bilinginformation | GovernmentiDNumber | Hobby | Atea | NewCall | NumberCats [~
Billmarity, on 15t of morth. 964093436 Mountg cinbing Cti 97171393 500,00 &M 0
[ Bi every o weeks 2800 Gardering County | 7/2/1393 100,00 PM 3
| T2087.9%7 Kriting,Seving | Ciy 2
| 25093487 Burring Oy | £/251593 300,00PM T
| 2305094 Fihing County | £/2/1359 T1.00,00 &M 3
[ Pays prompty 209090673 Gardening Country | 7/10/1339 200:00 PM| 2

| 204909856 Cly | 7/121393600,00 M 0.

Record: 1l ([T > (o1 loklof 7 e |





Access Tables and Fields

Selecting this command from the Add-ins menu generates a table containing the names of fields in all the tables of the current database, and some of their properties, and opens a report based on this table in print preview.

[image: image9.png]2 Microsoft Access - [Table and Field Names] _[ofx
| & el Edit view Tools window Help - =] x|
-8 O @@ w0% - cose| ¥ -0

|
Table and Field Names
Table Name ield Name Data Type Description  Required?  Validation Rule
thIContacts Address 10 Text No
City 10 Text No
CompanyName 10 Text No
ContactID 4 LongInteger  Yes
ContactTypeID 4 LongInteger  No
Country 10 Text No
Dear 10 Text No
EmailName 10 Text No
FaxNumber 10 Text No
FirstName 10 Text No
LastMectingDate 8 Date/Time No
LastName 10 Text No
MobilePhone 10 Text No
Notes 12 Memo No |
PostalCode 10 Text No
ReferredBy 10 Text No
Region 10 Text No
StateOrProvince 10 Text No
Title 10 Text No
WorkExtension 10 Text No
WorkPhone 10 Text No
P 1l 5 » In o I — o

[Ready. ) oy T 7




Code

Option Compare Database

Option Explicit

'Requires references to the Outlook and DAO object libraries

Dim strCallingDB As String

Dim strForm As String

Dim strCustomForm As String

Dim strCodeDB As String

Dim appOutlook As outlook.Application

Dim ins As outlook.Inspector

Dim exp As outlook.Explorer

Dim mapifld As outlook.MAPIFolder

'Just declared as Object because we don't know what type of item it is

Dim itm As Object

Dim dbsCalling As DAO.Database

Dim dbsCode As DAO.Database

Dim flds As DAO.Fields

Dim tdf As DAO.TableDef

Dim fld As DAO.Field

Dim prp As Object

Dim rst As DAO.Recordset

Dim rstProps As DAO.Recordset

Dim strTable As String

Dim strTableName As String

Dim strFieldName As String

Dim strSQL As String

Dim lngOutlookDataType As Long

Dim lngAccessDataType As Long

Dim strTitle As String

Dim strMessage As String

Dim intReturn As Integer

Dim fCheck As Boolean

Dim fStandard As Boolean

Public Function GetContactData()

'Written by Helen Feddema 12-16-1998

'Last modified 12-23-2001

On Error Resume Next

   'Determine whether Outlook is running

   Set appOutlook = GetObject(, "Outlook.Application")

   Debug.Print "GetObject Error #: " & Err.Number

   If Err.Number <> 0 Then

      'Outlook is not running; can't get a contact item to process

      MsgBox "Outlook is not running; can't import contacts"

      Exit Function

   End If

On Error GoTo GetContactDataError

   'Check that current folder is a contacts folder

   Set exp = appOutlook.ActiveExplorer

   Set mapifld = exp.CurrentFolder

   If mapifld.DefaultItemType <> 2 Then

      MsgBox "The active folder does not contain contact items; exiting"

      Exit Function

   End If

   Set appOutlook = Nothing

   Set exp = Nothing

   Set mapifld = Nothing

   Call OutlookCustomProperties

GetContactDataExit:

   Exit Function

GetContactDataError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume GetContactDataExit

End Function

Public Function OutlookCustomProperties()

'Written by Helen Feddema 12-4-1998

'Last modified 12-23-2001

On Error Resume Next

   Dim fCustomProps As Boolean

   Dim strQuery As String

   Dim strMatch As String

   Dim itms As Object

   Dim lngCount As Long

   strCodeDB = CodeDb.Name

   strCallingDB = CurrentDb.Name

   strTable = "tblOutlookContacts"

   'Close tblOutlookContacts if it is open

   DoCmd.Close acTable, strTable

   'Rename tblOutlookContacts if it exists

   DoCmd.SetWarnings False

   DoCmd.Rename "_tblOriginalContacts", acTable, strTable

On Error GoTo OutlookCustomPropertiesError

   'Copy table with standard field names to blank tblOutlookContacts

   Set dbsCalling = CurrentDb

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acTable, _

      Source:="zstblStandardContactProps", _

      Destination:=strTable, structureonly:=False

   'Determine item class and form type

   Set appOutlook = GetObject(, "Outlook.Application")

   Set exp = appOutlook.ActiveExplorer

   Set mapifld = exp.CurrentFolder

   Set ins = appOutlook.ActiveInspector

   Debug.Print "Current folder: " & mapifld

   Debug.Print "Current item class: " & ins.CurrentItem.Class

   If ins.CurrentItem.Class <> 40 Then

      'Current item is not a contact item

      strTitle = "Item Type"

      strMessage = "No contact item is open;"

      strMessage = strMessage & " do you want to process all "

      strMessage = strMessage & " the contact items in the " _

         & mapifld.Name & " folder using the standard Contact form?"

      intReturn = MsgBox(strMessage, vbYesNoCancel + _

         vbQuestion + vbDefaultButton1, strTitle)

      If intReturn = vbYes Then

         fCheck = False

         fStandard = True

         GoTo ImportStandardData

      Else

         Exit Function

      End If

   ElseIf ins.CurrentItem.Class = 40 Then

      'Current item is a contact item

      Set itm = ins.CurrentItem

      strForm = itm.MessageClass

      Debug.Print "Message class from form: " & strForm

      'Save strForm to table for later use

      Call SaveCustomFormName(strForm)

      strTitle = "Form Type"

      If strForm <> "IPM.Contact" Then

         'Current item is based on a custom form

         fStandard = False

         strCustomForm = Mid(strForm, 13)

         'Check whether there are any custom fields in the Outlook

         'item and go directly to CreateModule if so

         If itm.UserProperties.Count = 0 Then

            fCustomProps = False

            'MsgBox "No Outlook custom properties in this contact item"

            GoTo ImportStandardData

         Else

            fCustomProps = True

            GoTo ImportCustomData

         End If

         'Custom form does not have any custom fields

         strMessage = "Limit import to contact items based on the " _

            & strCustomForm & " form?"

         intReturn = MsgBox(strMessage, vbYesNoCancel + vbQuestion + _

            vbDefaultButton1, strTitle)

         If intReturn = vbNo Then

            fCheck = False

            GoTo ImportStandardData

         ElseIf intReturn = vbYes Then

            fCheck = True

            GoTo ImportStandardData

         ElseIf intReturn = vbCancel Then

            Exit Function

         End If

      ElseIf strForm = "IPM.Contact" Then

         'Current item is based on the standard Contact form

         fStandard = True

         strMessage = "Limit import to standard contact items?"

         intReturn = MsgBox(strMessage, vbYesNoCancel + vbQuestion + _

            vbDefaultButton1, strTitle)

         If intReturn = vbNo Then

            fCheck = False

            GoTo ImportStandardData

         ElseIf intReturn = vbYes Then

            fCheck = True

            GoTo ImportStandardData

         ElseIf intReturn = vbCancel Then

            GoTo OutlookCustomPropertiesError

         End If

      End If

   End If

ImportCustomData:

   Debug.Print "Contact item for " & itm.FullName

   Debug.Print itm.UserProperties.Count & " custom Outlook properties"

   strTable = "zstblCustomPropertyNames"

   'Delete old tables in calling database (if they exist)

On Error Resume Next

   dbsCalling.TableDefs.Delete strTable

   dbsCalling.TableDefs.Delete "zstblOAInfo"

On Error GoTo OutlookCustomPropertiesError

   'Generate table to hold field names

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acTable, Source:=strTable, _

      Destination:=strTable, structureonly:=True

   'Fill table with property names

   Set rst = dbsCalling.OpenRecordset(strTable, dbOpenTable)

   For Each prp In itm.UserProperties

      With rst

         .AddNew

         Debug.Print "Adding property: " & prp.Name

         !PropertyName = prp.Name

         !DataType = prp.Type

         .Update

      End With

   Next prp

   rst.Close

   'Append custom property names to tblOutlookContacts

   strTable = "tblOutlookContacts"

   Set tdf = dbsCalling.TableDefs(strTable)

   Debug.Print "Table name: " & tdf.Name

   For Each prp In itm.UserProperties

      strFieldName = prp.Name

      lngOutlookDataType = prp.Type

      'Convert Outlook data types to closest Access data types

      Select Case lngOutlookDataType

         Case 1

            lngAccessDataType = dbText

         Case 3

            lngAccessDataType = dbLong

         Case 5

            lngAccessDataType = dbDate

         Case 6

            lngAccessDataType = dbBoolean

         Case 7

            lngAccessDataType = dbLong

         Case 11

            lngAccessDataType = dbText

         Case 12

            lngAccessDataType = dbSingle

         Case 14

            lngAccessDataType = dbCurrency

         Case 18

            lngAccessDataType = dbText

         Case 19

            lngAccessDataType = dbText

      End Select

      Debug.Print "Outlook field name: " & strFieldName

      Debug.Print "Outlook data type: " & lngOutlookDataType

      Debug.Print "Access Data Type: " & lngAccessDataType

      Set fld = tdf.CreateField(strFieldName, lngAccessDataType)

      'fld.AllowZeroLength = True

      tdf.Fields.Append fld

   Next prp

   'Call function to write lines of code for the custom properties,

   'and create a separate module to import the contact data

   Call CreateModule(strForm, itm, fCheck)

   GoTo OutlookCustomPropertiesExit

ImportStandardData:

   If fCheck = True Then

      strMatch = "[MessageClass] = " & Chr$(34) & strForm & Chr$(34)

      Debug.Print "Restrict clause: " & strMatch

      Set itms = mapifld.Items.Restrict(strMatch)

      If fStandard = False Then

         strMessage = itms.Count & " contact items based on the " _

            & Mid(strForm, 13) & " form in the " & mapifld.Name & " folder"

      ElseIf fStandard = True Then

         strMessage = itms.Count & " contact items based on the " _

            & "standard Contact form in the " & mapifld.Name & " folder"

      End If

      MsgBox strMessage

   ElseIf fCheck = False Then

      Set itms = mapifld.Items

      strMessage = itms.Count & " contact items in the " _

         & mapifld.Name & " folder"

      MsgBox strMessage

   End If

   If itms.Count = 0 Then

      MsgBox "No contact items in the " & mapifld.Name _

         & " folder to process; exiting"

      Exit Function

   End If

   'Check whether tblOutlookContacts has any records and ask

   'if user wants to clear table or append to it

   lngCount = Nz(DCount("*", strTable))

   If lngCount > 0 Then

      strTitle = "Clear or Append?"

      strMessage = "Clear table or append new records to existing records?" & vbCrLf _

         & "(Yes clears; No appends)"

      intReturn = MsgBox(strMessage, vbYesNoCancel + _

         vbQuestion + vbDefaultButton1, strTitle)

      If intReturn = vbYes Then

         DoCmd.SetWarnings False

         strSQL = "DELETE * FROM tblOutlookContacts;"

         DoCmd.RunSQL strSQL

         GoTo FillTable

      ElseIf intReturn = vbNo Then

         GoTo FillTable

      ElseIf intReturn = vbCancel Then

         GoTo OutlookCustomPropertiesExit

      End If

   End If

FillTable:

   Set rst = dbsCalling.OpenRecordset(strTable, dbOpenTable)

   For Each itm In itms

      Debug.Print "Processing " & Nz(itm.FullName) & " item"

      With rst

         .AddNew

         'Standard Outlook fields

         !CustomerID = Nz(itm.CustomerID)

         !FullName = Nz(itm.FullName)

      Debug.Print "Full Name: " & Nz(itm.FullName)

         !Title = Nz(itm.Title)

         !FirstName = Nz(itm.FirstName)

         !MiddleName = Nz(itm.MiddleName)

         !LastName = Nz(itm.LastName)

         !Suffix = Nz(itm.Suffix)

         !NickName = Nz(itm.NickName)

         !CompanyName = Nz(itm.CompanyName)

         !Department = Nz(itm.Department)

         !JobTitle = Nz(itm.JobTitle)

         !BusinessAddress = Nz(itm.BusinessAddress)

         !BusinessAddressStreet = Nz(itm.BusinessAddressStreet)

         !BusinessAddressPostOfficeBox = Nz(itm.BusinessAddressPostOfficeBox)

         !BusinessAddressCity = Nz(itm.BusinessAddressCity)

         !BusinessAddressState = Nz(itm.BusinessAddressState)

         !BusinessAddressPostalCode = Nz(itm.BusinessAddressPostalCode)

         !BusinessAddressCountry = Nz(itm.BusinessAddressCountry)

         !BusinessHomePage = Nz(itm.BusinessHomePage)

         !ComputerNetworkName = Nz(itm.ComputerNetworkName)

         !FTPSite = Nz(itm.FTPSite)

         !HomeAddress = Nz(itm.HomeAddress)

         !HomeAddressStreet = Nz(itm.HomeAddressStreet)

         !HomeAddressPostOfficeBox = Nz(itm.HomeAddressPostOfficeBox)

         !HomeAddressCity = Nz(itm.HomeAddressCity)

         !HomeAddressState = Nz(itm.HomeAddressState)

         !HomeAddressPostalCode = Nz(itm.HomeAddressPostalCode)

         !HomeAddressCountry = Nz(itm.HomeAddressCountry)

         !OtherAddress = Nz(itm.OtherAddress)

         !OtherAddressStreet = Nz(itm.OtherAddressStreet)

         !OtherAddressPostOfficeBox = Nz(itm.OtherAddressPostOfficeBox)

         !OtherAddressCity = Nz(itm.OtherAddressCity)

         !OtherAddressState = Nz(itm.OtherAddressState)

         !OtherAddressPostalCode = Nz(itm.OtherAddressPostalCode)

         !OtherAddressCountry = Nz(itm.OtherAddressCountry)

         !MailingAddress = Nz(itm.MailingAddress)

         !AssistantTelephoneNumber = Nz(itm.AssistantTelephoneNumber)

         !BusinessFaxNumber = Nz(itm.BusinessFaxNumber)

         !BusinessTelephoneNumber = Nz(itm.BusinessTelephoneNumber)

         !Business2TelephoneNumber = Nz(itm.Business2TelephoneNumber)

         !CallbackTelephoneNumber = Nz(itm.CallbackTelephoneNumber)

         !CarTelephoneNumber = Nz(itm.CarTelephoneNumber)

         !CompanyMainTelephoneNumber = Nz(itm.CompanyMainTelephoneNumber)

         !HomeFaxNumber = Nz(itm.HomeFaxNumber)

         !HomeTelephoneNumber = Nz(itm.HomeTelephoneNumber)

         !Home2TelephoneNumber = Nz(itm.Home2TelephoneNumber)

         !ISDNNumber = Nz(itm.ISDNNumber)

         !MobileTelephoneNumber = Nz(itm.MobileTelephoneNumber)

         !OtherFaxNumber = Nz(itm.OtherFaxNumber)

         !OtherTelephoneNumber = Nz(itm.OtherTelephoneNumber)

         !PagerNumber = Nz(itm.PagerNumber)

         !PrimaryTelephoneNumber = Nz(itm.PrimaryTelephoneNumber)

         !RadioTelephoneNumber = Nz(itm.RadioTelephoneNumber)

         !TTYTDDTelephoneNumber = Nz(itm.TTYTDDTelephoneNumber)

         !TelexNumber = Nz(itm.TelexNumber)

         !Account = Nz(itm.Account)

         If itm.Anniversary <> "1/1/4501" Then

            !Anniversary = Nz(itm.Anniversary)

         End If

         !AssistantName = Nz(itm.AssistantName)

         !BillingInformation = Nz(itm.BillingInformation)

         If itm.Birthday <> "1/1/4501" Then

            !Birthday = Nz(itm.Birthday)

         End If

         !Categories = Nz(itm.Categories)

         !Children = Nz(itm.Children)

         !PersonalHomePage = Nz(itm.PersonalHomePage)

         !Email1Address = Nz(itm.Email1Address)

         !Email1DisplayName = Nz(itm.Email1DisplayName)

         !Email2Address = Nz(itm.Email2Address)

         !Email2DisplayName = Nz(itm.Email2DisplayName)

         !Email3Address = Nz(itm.Email3Address)

         !Email3DisplayName = Nz(itm.Email3DisplayName)

         !Gender = Switch(itm.Gender = 1, "Female", _

            itm.Gender = 2, "Male", itm.Gender = 0, "Unspecified")

         !GovernmentIDNumber = Nz(itm.GovernmentIDNumber)

         !Hobby = Nz(itm.Hobby)

         !Initials = Nz(itm.Initials)

         !Language = Nz(itm.Language)

         !ManagerName = Nz(itm.ManagerName)

         !Body = Nz(itm.Body)

         !OfficeLocation = Nz(itm.OfficeLocation)

         !OrganizationalIDNumber = Nz(itm.OrganizationalIDNumber)

         !Profession = Nz(itm.Profession)

         !ReferredBy = Nz(itm.ReferredBy)

         !Sensitivity = Switch(itm.Sensitivity = 3, "Confidential", _

         itm.Sensitivity = 0, "Normal", itm.Sensitivity = 1, _

         "Personal", itm.Sensitivity = 2, "Private")

         !Spouse = Nz(itm.Spouse)

         !User1 = Nz(itm.User1)

         !User2 = Nz(itm.User2)

         !User3 = Nz(itm.User3)

         !User4 = Nz(itm.User4)

         !WebPage = Nz(itm.WebPage)

         .Update

      End With

   Next itm

   rst.Close

   If fCheck = True Then

      If fStandard = True Then

         strMessage = "All contact items based on the standard " _

            & "contact form in the " & mapifld.Name _

            & " folder imported!"

      End If

   ElseIf fCheck = False Then

      strMessage = "All contacts in the " & mapifld.Name _

         & " folder imported!"

   End If

   MsgBox strMessage

   DoCmd.OpenTable strTable

OutlookCustomPropertiesExit:

   Exit Function

OutlookCustomPropertiesError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume OutlookCustomPropertiesExit

End Function

Public Function CreateModule(strForm As String, itm As Object, _

   fCheck As Boolean)

'Creates a module with a function to import data from a custom Outlook contact form

'with custom fields

'Written by Helen Feddema 5-18-1999

'Last modified 12-25-2001

On Error Resume Next

   Dim strMatch As String

   Dim strTextFile1 As String

   Dim strTextFile2 As String

   Dim strOfficeDir As String

   Dim mdl As Access.Module

   Dim strModuleName As String

   Dim strLine As String

   Dim strCodeLines As String

   Dim strModuleBeginning As String

   Dim strModuleEnd As String

   strCodeDB = CodeDb.Name

   strCallingDB = CurrentDb.Name

   fStandard = False

   'Delete existing modules and macro in calling database (if they exist)

   DoCmd.DeleteObject acModule, "basGetCustomContactData"

   DoCmd.DeleteObject acMacro, "mcrImportCustomContactData"

On Error GoTo ErrorHandler

   'Copy macro to calling database

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acMacro, _

      Source:="zsmcrImportCustomContactData", _

      Destination:="mcrImportCustomContactData"

   'Copy table with custom form name to calling database

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acTable, _

      Source:="zstblOAInfo", Destination:="zstblOAInfo", _

      structureonly:=False

   'Retrieve stored module beginning and module end text from table

   Set dbsCode = CodeDb

   Set rst = dbsCode.OpenRecordset("zstblInfo", dbOpenTable)

   With rst

      .MoveFirst

      strModuleBeginning = ![ModuleBeginning]

      strModuleEnd = ![ModuleEnd]

      .Close

   End With

   'Create string with lines of code for custom Outlook field names

   Set dbsCalling = CurrentDb

   Set rstProps = dbsCalling.OpenRecordset("zstblCustomPropertyNames")

   Debug.Print "Number of custom fields: " & rstProps.RecordCount

   Do Until rstProps.EOF

      strFieldName = rstProps!PropertyName

      lngAccessDataType = rstProps!DataType

      Debug.Print "Custom field name: " & strFieldName & "; data type: " & lngAccessDataType

      Select Case lngAccessDataType

         'Treat values differently according to their Outlook data type

         Case 1

            'Text

            strLine = "         ![" & strFieldName & "] = " & _

               "Nz(itm.UserProperties(" & Chr$(34) & strFieldName & Chr$(34) & "))"

            Debug.Print "Code line: " & strLine

         Case 3, 12

            'Number

            strLine = "         ![" & strFieldName & "] = " _

               & "Nz(itm.UserProperties(" & Chr$(34) & strFieldName & Chr$(34) & "))"

            Debug.Print "Code line: " & strLine

         Case 5

            'DateTime

            strLine = "         If Nz(itm.UserProperties(" & Chr$(34) & strFieldName _

               & Chr$(34) & ")) <> " & Chr(34) & "1/1/4501" & Chr(34) & "Then" & vbCrLf _

               & "            ![" & strFieldName & "] = " & "         Nz(itm.UserProperties(" _

               & Chr$(34) & strFieldName & Chr$(34) & "))" & vbCrLf _

               & "         End If"

               Debug.Print "Code line: " & strLine

         Case Else

            strLine = "         ![" & strFieldName & "] = " _

               & "Nz(itm.UserProperties(" & Chr$(34) & strFieldName & Chr$(34) & "))"

            Debug.Print "Code line: " & strLine

      End Select

      strCodeLines = IIf(Nz(strCodeLines) <> "", strCodeLines & _

         vbCrLf & strLine, strLine)

      rstProps.MoveNext

   Loop

   rstProps.Close

   'Create module in calling database to fill with custom contact data

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acModule, _

      Source:="zsbasGetCustomContactData", _

      Destination:="basGetCustomContactData"

   'Combine three parts of module into a single string

   strCodeLines = strModuleBeginning & vbCrLf & strCodeLines _

      & vbCrLf & strModuleEnd

   'Paste text into module

   strModuleName = "basGetCustomContactData"

   DoCmd.OpenModule strModuleName

   Set mdl = Modules(strModuleName)

   mdl.AddFromString strCodeLines

   DoCmd.Close acModule, strModuleName

   strMessage = "Run the mcrImportCustomContactData macro whenever" _

      & vbCrLf & "you want to fill the tblOutlookContacts table with data from" _

      & " the " & vbCrLf & strCustomForm & " custom Outlook form"

   MsgBox strMessage

   'Close VBE window

   SendKeys "%{F4}"

ErrorHandlerExit:

   Exit Function

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume ErrorHandlerExit

End Function

Public Function AccessTablesAndFields()

'Written by Helen Feddema 6-17-1998

'Last modified 12-23-2001

On Error Resume Next

   Dim strReport As String

   strReport = "zsrptTableAndFieldNames"

   strCodeDB = CodeDb.Name

   'Delete old table in calling database (if it exists)

   strTable = "zstblAccessDataTypes"

   Set dbsCalling = CurrentDb

   dbsCalling.TableDefs.Delete strTable

On Error GoTo AccessTablesAndFieldsError

   'Copy table to calling database

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acTable, Source:=strTable, _

      Destination:=strTable, structureonly:=False

   'Delete old table in calling database (if it exists)

   strTable = "zstblTableAndFieldNames"

On Error Resume Next

   dbsCalling.TableDefs.Delete strTable

On Error GoTo AccessTablesAndFieldsError

   'Copy table to calling database

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acTable, Source:=strTable, _

      Destination:=strTable, structureonly:=True

   'Delete old report in calling database (if it exists)

   strSQL = "SELECT MSysObjects.Name FROM MSysObjects WHERE "

   strSQL = strSQL & "MSysObjects.Name Like " _

      & Chr$(34) & strReport & Chr$(34) & ";"

   Debug.Print strSQL

   Set rst = dbsCalling.OpenRecordset(strSQL)

   If rst.RecordCount > 0 Then

      DoCmd.DeleteObject acReport, strReport

   End If

   rst.Close

   'Copy report to calling database

   DoCmd.TransferDatabase acImport, databasetype:="Microsoft Access", _

      databasename:=strCodeDB, objecttype:=acReport, Source:="rptTableAndFieldNames", _

      Destination:=strReport

   'Fill table with field names

   Set rst = dbsCalling.OpenRecordset(strTable, dbOpenTable)

   For Each tdf In dbsCalling.TableDefs

      strTableName = tdf.Name

      If left(strTableName, 4) <> "MSys" Then

         Set flds = tdf.Fields

         For Each fld In flds

            strFieldName = fld.Name

            With rst

               .AddNew

               !TableName = strTableName

               !FieldName = strFieldName

               !DataType = fld.Type

               !ValidationRule = fld.ValidationRule

               !Required = fld.Required

               .Update

            End With

         Next fld

      End If

   Next tdf

   rst.Close

   'dbs.Close

   DoCmd.OpenReport strReport, acViewPreview

AccessTablesAndFieldsExit:

   Exit Function

AccessTablesAndFieldsError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume AccessTablesAndFieldsExit

End Function

Public Function oaStripNonAlphaNumericChars(strText As String) As String

'Strips a variety of non-alphanumeric characters from a text string

'Created by Helen Feddema 10-15-97

'Modified by Ruud H.G. van Tol 6-18-99

'Last modified by Helen Feddema 7-2-99

On Error GoTo ErrorHandler

   Dim strTestString As String

   Dim strTestChar As String

   Dim lngFound As Long

   Dim i As Integer

   Dim strStripChars As String

   strStripChars = " `~!@#$%^&*()-_=+[{]};:',<.>/?" & Chr$(34) & Chr$(13) & Chr$(10)

   strTestString = strText

   i = 1

   Do While i <= Len(strTestString)

     'Find a strippable character

     strTestChar = Mid$(strTestString, i, 1)

     lngFound = InStr(strStripChars, strTestChar)

     If lngFound > 0 Then

       strTestString = left(strTestString, i - 1) & Mid(strTestString, i + 1)

     Else

       i = i + 1

     End If

   Loop

   oaStripNonAlphaNumericChars = strTestString

ErrorHandlerExit:

   Exit Function

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " _

      & Err.Description

   Resume ErrorHandlerExit

End Function

Sub SaveCustomFormName(strForm As String)

'Created by Helen Feddema 12-23-2001

'Last modified 12-23-2001

On Error GoTo ErrorHandler

   'Save custom form name to table

   Set dbsCode = CodeDb

   Set rst = dbsCode.OpenRecordset("zstblOAInfo")

   With rst

      .MoveFirst

      .Edit

      ![FormName] = strForm

      .Update

      .Close

   End With

ErrorHandlerExit:

   Exit Sub

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume ErrorHandlerExit

End Sub

Notes

· Because of a difference in the way OpenModule works in Access 2000, I had to modify some of the CreateModule code from the Access 97 version, which left the VBE window open at the end of the CreateModule function.  I put in a SendKeys command to close it.

· For full details on the design of the Access 97 version of this add-in, see my article in the July 1999 issue of Smart Access.  Smart Access is published by Pinnacle, http://www.pinpub.com.

· If you get the error message shown below when running the macro or code to do an import from custom contact items, you are missing one (or both) of the required references; see Step 5 in the Get Contact Data section above for details on the references required by this add-in. 

Contents of Zip File

	File Name
	File Type
	Destination Folder

	Outlook Automation.mda
	Access 2002 add-in
	Add-ins folder (usually C:\WINDOWS\Application Data\Microsoft\AddIns for Windows ME or C:\Documents and Settings\UserName\Application Data\Microsoft\AddIns for Windows 2000)

	OA Test Form.oft
	Outlook form template
	\Templates\Outlook

	*.msg
	Outlook custom Contact items
	\My Documents\ (or wherever you want)

Drag them to a Contacts-type folder to test exporting custom contact items

	VBA Code Sample #38 – The Outlook Automation Access 2002 Add-in.doc
	Word document
	\My Documents\ (or wherever you want)

